# Gauss principle for (exact) constant temperature

This is an illustration of Gauss' principle for the derivation of the equations of motion for a harmonic oscillator at **constant kinetic energy**.

## Non-holonomic constraint

### Definition

Constant kinetic energy

In[1]:=
```
σ = m (x'[t]^2 + y'[t]^2 + z'[t]^2) / 2 == 3 k_B T / 2
```

Out[1]= $\frac{1}{2} m \left(x'[t]^2 + y'[t]^2 + z'[t]^2\right) == \frac{3 T k_B}{2}$

### Resulting acceleration constraint

In[2]:=
```
Aconstr = D[σ, t] // Simplify
```

Out[2]= $m\,(x'[t]\, x''[t] + y'[t]\, y''[t] + z'[t]\, z''[t]) == 0$

Corresponding constraint function fulfilling $g[\ddot{r}] \equiv 0$

In[3]:=
```
g = Aconstr[[1]]
```

Out[3]= $m\,(x'[t]\, x''[t] + y'[t]\, y''[t] + z'[t]\, z''[t])$

## Q-function

Gravitational force

In[4]:=
```
F = -K * {x[t], y[t], z[t]}
```

Out[4]= $\{-K\,x[t], -K\,y[t], -K\,z[t]\}$

Acceleration

In[5]:=
```
a = {x''[t], y''[t], z''[t]}
```

Out[5]= $\{x''[t], y''[t], z''[t]\}$

Extended Q function

In[6]:= `Qext = m / 2 (a - F / m) . (a - F / m) + μ * g`

Out[6]= $$m\,\mu\,(x'[t]\,x''[t]+y'[t]\,y''[t]+z'[t]\,z''[t])+\frac{1}{2}m\left(\left(\frac{K\,x[t]}{m}+x''[t]\right)^2+\left(\frac{K\,y[t]}{m}+y''[t]\right)^2+\left(\frac{K\,z[t]}{m}+z''[t]\right)^2\right)$$

# Minimize Q-function

Gradient of Q with respect to the accelerations

In[7]:= `MatrixForm[GradQ = Grad[Qext, a] // Simplify]`

Out[7]//MatrixForm=
$$\begin{pmatrix} K\,x[t]+m\,(\mu\,x'[t]+x''[t]) \\ K\,y[t]+m\,(\mu\,y'[t]+y''[t]) \\ K\,z[t]+m\,(\mu\,z'[t]+z''[t]) \end{pmatrix}$$

General form of the accelerations

In[8]:= `asol = Table[Solve[GradQ[[i]] == 0, a[[i]]], {i, 1, 3}]`

Out[8]= $$\left\{\left\{\left\{x''[t]\to\frac{-K\,x[t]-m\,\mu\,x'[t]}{m}\right\}\right\},\left\{\left\{y''[t]\to\frac{-K\,y[t]-m\,\mu\,y'[t]}{m}\right\}\right\},\left\{\left\{z''[t]\to\frac{-K\,z[t]-m\,\mu\,z'[t]}{m}\right\}\right\}\right\}$$

In[9]:= `aGen = Table[Expand[asol[[i, 1, 1, 2]]], {i, 1, 3}]`

Out[9]= $$\left\{-\frac{K\,x[t]}{m}-\mu\,x'[t],\,-\frac{K\,y[t]}{m}-\mu\,y'[t],\,-\frac{K\,z[t]}{m}-\mu\,z'[t]\right\}$$

# Compute Lagrange multipliers

In[10]:= `ReplacementRulesA = asol[[All, 1, 1]]`

Out[10]= $$\left\{x''[t]\to\frac{-K\,x[t]-m\,\mu\,x'[t]}{m},\,y''[t]\to\frac{-K\,y[t]-m\,\mu\,y'[t]}{m},\,z''[t]\to\frac{-K\,z[t]-m\,\mu\,z'[t]}{m}\right\}$$

In[11]:= `Eq = (g == 0) /. ReplacementRulesA`

Out[11]= $$m\left(\frac{x'[t]\,(-K\,x[t]-m\,\mu\,x'[t])}{m}+\frac{y'[t]\,(-K\,y[t]-m\,\mu\,y'[t])}{m}+\frac{z'[t]\,(-K\,z[t]-m\,\mu\,z'[t])}{m}\right)==0$$

In[12]:= `μsol = FullSimplify[Solve[Eq, μ], Assumptions → {σ}]`

Out[12]= $$\left\{\left\{\mu\to-\frac{K\,(x[t]\,x'[t]+y[t]\,y'[t]+z[t]\,z'[t])}{3\,T\,k_B}\right\}\right\}$$

# Final equation of motion

In[13]:= `ReplacementRulesMu = μsol[[1, All]]`

Out[13]= $$\left\{\mu \to -\frac{K\,(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t])}{3\,T\,k_B}\right\}$$

In[14]:= `aGen`

Out[14]= $$\left\{-\frac{K\,x[t]}{m} - \mu\,x'[t],\ -\frac{K\,y[t]}{m} - \mu\,y'[t],\ -\frac{K\,z[t]}{m} - \mu\,z'[t]\right\}$$

In[15]:= `aSol = FullSimplify[aGen /. ReplacementRulesMu, Assumptions → {σ}]`

Out[15]= $$\left\{-\frac{K\,x[t]}{m} + \frac{K\,x'[t]\,(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t])}{3\,T\,k_B},\ -\frac{K\,y[t]}{m} + \frac{K\,y'[t]\,(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t])}{3\,T\,k_B},\ -\frac{K\,z[t]}{m} + \frac{K\,z'[t]\,(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t])}{3\,T\,k_B}\right\}$$

In[16]:= `EqMotGauss = Table[a[[i]] == aSol[[i]], {i, 1, 3}];`

In[17]:= `MatrixForm[EqMotGauss]`

Out[17]//MatrixForm=

$$\begin{pmatrix} x''[t] == -\frac{K\,x[t]}{m} + \frac{K\,x'[t]\,(x[t]\,x'[t]+y[t]\,y'[t]+z[t]\,z'[t])}{3\,T\,k_B} \\ y''[t] == -\frac{K\,y[t]}{m} + \frac{K\,y'[t]\,(x[t]\,x'[t]+y[t]\,y'[t]+z[t]\,z'[t])}{3\,T\,k_B} \\ z''[t] == -\frac{K\,z[t]}{m} + \frac{K\,z'[t]\,(x[t]\,x'[t]+y[t]\,y'[t]+z[t]\,z'[t])}{3\,T\,k_B} \end{pmatrix}$$